POLSKA AKADEMIA NAUK
Instytut Badań Systemowych

ROZWÓJ I ZASTOSOWANIA TECHNOLOGII I SYSTEMÓW INFORMATYCZNYCH

pod redakcją:
Jana Studzińskiego
Ludosława Drelichowskiego
Olgierda Hryniewicza

ROZWÓJ I ZASTOSOWANIA TECHNOLOGII
I SYSTEMÓW INFORMATYCZNYCH

Polska Akademia Nauk • Instytut Badań Systemowych

Seria: BADANIA SYSTEMOWE
tom 28

Redaktor naukowy:

Prof. dr hab. Jakub Gutenbaum

Warszawa 2001

ROZWÓJ I ZASTOSOWANIA TECHNOLOGII I SYSTEMÓW INFORMATYCZNYCH

pod redakcją
Jana Studzińskiego, Ludosława Drelichowskiego
i Olgierda Hryniewicza

Wydano z wykorzystaniem dotacji KOMITETU BADAŃ NAUKOWYCH

Książka zawiera wybór artykułów poświęconych omówieniu aktualnego stanu badań w kraju w zakresie rozwoju technologii, modeli i systemów informatycznych oraz ich zastosowań w różnych dziedzinach gospodarki narodowej. Wyodrębnioną grupę stanowią artykuły aplikacyjne omawiające wyniki projektów badawczych i celowych KBN.

Recenzenci artykułów:

Dr hab. inż. Ryszard Budziński, prof. US
Prof. dr hab. inż. Janusz Kacprzyk
Dr hab. Adam Kopiński, prof. AE we Wrocławiu
Doc dr hab. inż. Marek Libura
Prof. dr hab. inż. Andrzej Straszak

ISBN 83-85847-59-6
ISSN 0208-8028

Rozdział 2

Metodologia i rozwój systemów informatycznych

MODELOWANIE I OPTYMALIZACJA SYSTEMÓW PRZY POMOCY UNIWERSALNYCH NARZĘDZI INFORMATYCZNYCH

Marek Miłosz
Katedra Informatyki, Politechnika Lubelska
marekm@pluton.pol.lublin.pl

W artykule przedstawione zostały problemy budowy i rozwiązywania modeli matematycznych dużych systemów gospodarczych. Zwrócono uwagę na możliwość efektywnego wykorzystania uniwersalnych narzędzi informatycznych zintegrowanych przy pomocy standardowych technik wymiany danych pomiędzy różnymi programami. Przedstawiono przykład takiej architektury, opracowanej dla modelu komputerowego optymalizacji struktury parku samochodowego w ujęciu gałęziowym.

1. Wstęp

W trakcie prac studialnych, naukowo-badawczych i prototypowych stosowane są modele matematyczne systemów. Modele takie mają zwykle dużą złożoność. Złożoność tą wyznacza liczba równań tworzących model, ale też i liczba zmiennych stanu, zmiennych decyzyjnych, parametrów, ograniczeń itp. By wykorzystać modele w praktyce niezbędna jest ich konkretyzacja, tj. wyznaczenie wartości wszystkich współczynników modelu. Wartości te są pozyskiwane z systemów rzeczywistych i przechowywane w bazie danych. Komputerowa implementacja modelu powinna być zatem zintegrowana ze zbiorem jego współczynników.

Z drugiej strony modelowanie zwykle powiązane jest z rozwiązaniem zagadnień optymalizacyjnych, czyli takich, które umożliwiają wybór najlepszego z wariantów. Jest to klasyczna metodyka wykorzystania analizy systemowej do badań systemów rzeczywistych (rys. 1).

W trakcie badań realnych systemów rzadko udaje się poprzestać na jednorodnym, pojedynczym modelu. Badacz wykorzystuje różne modele i, w konsekwencji, zmuszony jest do użycia wielu narzędzi informatycznych.

Proces komputeryzacji modeli matematycznych można realizować na dwa sposoby: poprzez tworzenie specjalnego oprogramowania w wybranym uniwersalnym języku programowania lub poprzez wykorzystanie gotowych narzędzi i ich integrację dla osiągnięcia celów badawczych.



Rys. 1. Klasyczne podejście do modelowania w analizie systemowej (Findeisen, 1985)

2. Uniwersalność a specjalizacja

Wykorzystanie uniwersalnych języków programowania do budowy modeli komputerowych złożonych systemów ma wiele zalet, ale i wad. Programy tworzone specjalnie do rozwiązania konkretnych problemów działają szybko (są zwykle bardzo efektywne, bo specjalnie stworzone do danego problemu). Z drugiej strony podejście to posiada szereg wad:

- Trwa długo i nakłady pracy są zwykle bardzo duże.
- Oprogramowanie jest mało elastyczne – specjalizowane, co jest istotną wadą w przypadku modelowania systemów nowych, a z takimi ma się zwykle do czynienia w pracach naukowo-badawczych.
- W trakcie prac nad stworzeniem systemu znaczna część wysiłków jest ukierunkowana na powtarzanie już wcześniej wykonanych przez kogoś prac –„odkrywanie" starych algorytmów i ich ponowną realizacje.
- Wydłużony i czasochłonny proces testowania zrealizowanych modeli, nie tylko pod kątem poprawności modelu, ale też podstawowych procedur i funkcji.

Uniwersalne narzędzia informatyczne oferują badaczowi następujące możliwości:

- Szybkie i elastyczne (podatne na zmiany) konstruowanie modeli komputerowych systemów.
- Łatwość integracji z innymi modelami i bazami danych statystycznych i normatywnych modeli.
- Brak konieczności testowania oprogramowania narzędziowego – powoduje to możliwość koncentracji na zadaniach merytorycznych a nie narzędziowych.

Można więc zaryzykować stwierdzenie, że w zadaniach jednostkowych, naukowo-badawczych czy prototypach uniwersalne systemy informatyczne są idealnymi narzędziami.

3. Środowisko budowy modeli systemu

Zespół Politechniki Lubelskiej i Instytutu Transportu Samochodowego realizuje grant KBN „Narzędzia wspomagające kształtowanie polityki restrukturyzacji parku samochodowego" (Miłosz, 2000a). W ramach grantu powstał model optymalizacyjny o dużym rozmiarze (Miłosz, 2000b), połączony z modelem techniczno-ekonomicznym wyznaczającym współczynniki modelu. Do komputerowej implementacji modelu wybrano dwa uniwersalne narzędzia i dokonano ich integracji przy pomocy standardowych mechanizmów wymiany danych systemu MS Windows (OLE i ODBC): Excel i Lingo (Miłosz, Muryjas , 2000). Systemy te połączono w jeden spójny komputerowy model do badań strategii restrukturyzacji parku samochodów (rys. 2).

Przy pomocy arkusza kalkulacyjnego Excel zorganizowano bazę danych modelu systemu (jest to zwykła kartotekowa baza danych współczynników i parametrów wejściowych modelu) oraz model techniczno-ekonomiczny samochodu (plik KOSZT.xls – rys. 2). Model ten jest układem równań wyznaczających podstawowe parametry funkcjonowania samochodu w parku.

By automatyzować proces generowania bazy danych współczynników modelu symulacyjnego opracowany został program w VBA, który na podstawie parametrów wejściowych i modelu techniczno-ekonomicznego generuje podstawowe parametru modelu optymalizacyjnego.

Model ten z kolei jest rozwiązywany (tj. znajdowane jest optimum funkcji celu) przez uniwersalny solver Lingo (Miłosz, Muryjas, 2000). Model problemu optymalizacyjnego jest zintegrowany z danymi przechowywanymi w zeszycie PARK.xls (rys. 2) poprzez technologię OLE. Arkusz ten jest także bazą danych - rezultatów modelowania parku samochodów.

Model optymalizacyjny został opracowany w specjalnym języku sytemu Lingo i jest umieszczony w wejściowym pliku tekstowym PARK.lg4 (rys. 2 i listing 1). W pliku tym zawarte są także elementy integracji modeli (listing 2).

Rys. 2. Architektura systemu do analizy wpływu polityki państwa na park samochodów



```
MODEL:
      SETS:
            CZAS                  : ;
            MWIEK                 : ;
            POJAZD        : WIEK, WARTOSC_K;
            ZADANIE       : ;
            PxZ (ZADANIE, POJAZD) : PRZYPISZ ;
            CxP (CZAS, POJAZD)  : KSTALE, KZAKUPU, SUMA_T, SUMA_TW,
                                             ZAKUP,  STAN_P ;
            CxSP (MWIEK, POJAZD): STARY_P, CZAS_R, ZMNIEJSZENIE, KSPRZEDAZ;
            CxZxP(CZAS,ZADANIE,POJAZD): KZMIENNE, BILANS_ZAD, UDZIAL;
            CxZ (CZAS, ZADANIE)        : POTRZEBY;
      ENDSETS
! wielkość parku ;
      @FOR(POJAZD(I):   @FOR(CxP(T,I):
      STAN_P(T,I) =
            @SUM(CxSP(TAU, I)| TAU #LE# WIEK(I)-T+1 :
                  STARY_P(TAU,I)*ZMNIEJSZENIE(T+TAU,I) )
            + @SUM(CxP(TAU,I)| TAU#LE#T #AND# TAU#GT#@SMAX(T-WIEK(I),0)
                        :ZAKUP(TAU,I)*ZMNIEJSZENIE(T-TAU+1,I))));
! budżet roczny czasu pracy samochodów parku ;
      @FOR(POJAZD(I):   @FOR(CxP(T,I):
      SUMA_T(T,I) = @SUM(CxSP(TAU, I)| TAU #LE# @SMAX(WIEK(I)-T+1,0):
                  (STARY_P(TAU,I)*ZMNIEJSZENIE(T+TAU,I) *CZAS_R(TAU,I)))
            + @SUM(CxP(TAU,I)| TAU#LE#T #AND# TAU#GT#@SMAX(T-WIEK(I),0)
            :ZAKUP(TAU,I)*ZMNIEJSZENIE(T-TAU+1, I)*CZAS_R(T-TAU+1,I))));
! wartość końcowa samochodów parku ;
      @FOR(POJAZD(I):
      WARTOSC_K(I) = @SUM(CxSP(TAU, I)| TAU #LE# @SMAX(WIEK(I)-OKRES,0):
      (STARY_P(TAU,I)*ZMNIEJSZENIE(OKRES+TAU,I) *KSPRZEDAZ(TAU,I)))
                        + @SUM(CxP(TAU,I)| TAU#LE#OKRES #AND# TAU#GT#@SMAX(OKRES-
                        WIEK(I),0):
      ZAKUP(TAU,I)*ZMNIEJSZENIE(OKRES-TAU+1, I)*KSPRZEDAZ(OKRES-TAU+1,I)));
! budżet roczny czasu pracy samochodów parku - wykorzystany  ;
      @FOR(CZAS(T): @FOR(POJAZD(I):
                  SUMA_TW (T, I) = @SUM(CxZxP(T,J,I):
      PRZYPISZ(J,I)*BILANS_ZAD(T,J,I)*UDZIAL(T,J,I)*POTRZEBY(T,J)))) ;

! funkcja celu ;
MIN = @SUM(CxP(T,I): KZAKUPU(T,I)*ZAKUP(T,I) *@FPL(STOPA,T) ) +
      @SUM(CxP(T,I): KSTALE(T,I)*STAN_P(T,I)*@FPL(STOPA,T) ) +
            @SUM(CxZxP(T,J,I): KZMIENNE(T,J,I)* UDZIAL(T,J,I)*POTRZEBY(T,J)*@FPL(STOPA,T)
            )-
            @FPL(STOPA,OKRES)* @SUM(POJAZD(I): WARTOSC_K(I));

! warunek zbilansowania czasu pracy pojazdu - OK  ;
      @FOR(CZAS(T): @FOR(POJAZD(I): SUMA_TW (T, I) <= SUMA_T(T,I))) ;
! warunek konieczności wykonania każdego zadania - OK  ;
      @FOR(CZAS(T): @FOR(ZADANIE(J): @SUM(CxZxP(T,J,I): UDZIAL(T,J,I)*PRZYPISZ(J,I)) = 1));
```

Listing 1. Program-model optymalizacyjny (Miłosz, 2000a) w języku systemu Lingo

```
      DATA:
! wejścia ;
!       proste;
!              stopa procentowa   ;
             STOPA = @OLE('c:\its\PARK.XLS') ;
!              czasookres prognozy   ;
             OKRES = @OLE('c:\its\PARK.XLS') ;
!       zbiory;
!                    T - czas  ;
             CZAS = @OLE('c:\its\PARK.XLS') ;
!                    i - nr grupy pojazdów  ;
             POJAZD = @OLE('c:\its\PARK.XLS') ;
!                    max wiek rozpatrywany (roboczy)  ;
             MWIEK = @OLE('c:\its\PARK.XLS') ;
!                    stary park - ilości w grupach wiekowych ;
             STARY_P = @OLE('c:\its\PARK.XLS') ;
!                    wartość końcowa pojazdu  ;
             KSPRZEDAZ = @OLE('c:\its\PARK.XLS') ;
!                    max wiek pojazdu  ;
             WIEK = @OLE('c:\its\PARK.XLS') ;
!                    koszty stałe pojazdu  ;
             KSTALE = @OLE('c:\its\PARK.XLS') ;
!                    koszty zakupu nowego pojazdu  ;
             KZAKUPU = @OLE('c:\its\PARK.XLS') ;
!                    j - nr zadania  ;
             ZADANIE = @OLE('c:\its\PARK.XLS') ;
!                    wielkość potrzeb w j.nat. ;
             POTRZEBY = @OLE('c:\its\PARK.XLS') ;
!                    macierz możliwości realizacji 0/1 ;
             PRZYPISZ = @OLE('c:\its\PARK.XLS') ;
!                    roczny budżet czasu pracy pojazdu  ;
             CZAS_R = @OLE('c:\its\PARK.XLS');
!                    zmniejszenie liczby pojazdów w naturalny sposób  ;
             ZMNIEJSZENIE = @OLE('c:\its\PARK.XLS');
!                    koszty zmienne zł/zadanie  ;
             KZMIENNE = @OLE('c:\its\PARK.XLS') ;
!                    czas na zadanie  ;
             BILANS_ZAD = @OLE('c:\its\PARK.XLS') ;
!  wyjścia ;
!                    zakupy - zmienne decyzyjne  ;
             @OLE('c:\its\PARK.XLS', 'ZAKUP') = ZAKUP ;
!                    udział pojazdów z grupy w zadaniu  ;
             @OLE('c:\its\PARK.XLS', 'UDZIAL') = UDZIAL ;
!                    stan parku pojazdów  ;
             @OLE('c:\its\PARK.XLS', 'STAN_P') = STAN_P ;
!                    roczny budzet czasu pracy ;
             @OLE('c:\its\PARK.XLS', 'SUMA_T') = SUMA_T ;
!                    roczny wykorzystany budzet czasu pracy ;
             @OLE('c:\its\PARK.XLS', 'SUMA_TW') = SUMA_TW ;
!                    wartość końcowa pojazdów ;
             @OLE('c:\its\PARK.XLS', 'WARTOSC_K') = WARTOSC_K ;
      ENDDATA
```

Listing 2. Integracja danych modelu optymalizacyjnego z bazą danych w Excelu

Solver Lingo (rys. 3 i 4) umożliwia rozwiązywaniu bardzo dużych zadań optymalizacyjnych w rozsądnym czasie, z wykorzystaniem standardowych zasobów sprzętowych. Technologia OLE zapewnia przeniesienie rezultatów optymalizacji i późniejszego ich wykorzystania (tj. obróbki i przetwarzania) w środowisku systemu Excel. Możliwe jest zatem wykorzystanie standardowych narzędzi analitycznych i graficznych arkusza Excel.

Rys. 3. Model PARK w oknie solvera Lingo



Rys. 4. Raport rozwiązania modelu PARK – przykład dla zadania z ponad 400 zmiennych decyzyjnych

4. Podsumowanie

Praktyka pokazała, że uniwersalne narzędzi informatyczne (takie jak: arkusze kalkulacyjne i solvery) są doskonałymi narzędziami do modelowania i optymalizacji złożonych systemów. Zwalniają one badacza z kłopotliwych, bardzo pracochłonnych i dość ograniczonych funkcjonalnie prac programistycznych pozwalając skoncentrować się mu na problemach merytorycznych. Wobec doskonałych narzędzi integracyjnych połączenie wielu modeli w jedną spójną całość nie przedstawia problemu. Podobnie, dzięki makropoleceniom lub VBA, może być rozwiązany problem automatyzacji przetwarzań wykraczających poza standardowe funkcje tych programów.

Literatura

Findeisen W. (pod red.) (1985) *Analiza systemowa - podstawy i metodologia*. Praca zbiorowa. PWN, Warszawa,.

Miłosz M. (2000) *Model matematyczny dynamiki struktury parku samochodowego gałęzi transportu*. Instytut Badań Systemowych PAN, Seria: Badania systemowe, t. 26, str.: 228-241. (Technologie informatyczne w zarządzaniu. Systemy wspomagania decyzji. Pod red.: J.Studzińskiego, L.Drelichowskiego, O.Hryniewicza, J.Kacprzyka).

Miłosz M. (2000) *Modelowanie matematyczne jako element wspomagania polityki transportowej państwa*. W „Zarządzanie przedsiębiorstwem w otoczeniu rynkowym". pod red. W.Sitko, LTN, Kazimierz Dolny, str. 221-227. ISBN 83-87833-18-5.

Miłosz M., Muryjas P. (2000) *Uniwersalne narzędzia rozwiązywania problemów optymalizacyjnych*. IV Lubelskie Akademickie Forum Informatyczne. Kazimierz Dolny, 11-12.05., str. 123-132.

ISSN 0208-8028
ISBN 83-85847-59-6

W celu uzyskania bliższych informacji i zakupu dodatkowych egzemplarzy prosimy o kontakt z Instytutem Badań Systemowych PAN
ul. Newelska 6, 01-447 Warszawa
tel. 837-35-78 w. 241 e-mail: bibliote@ibspan.waw.pl